Stage IREM de LILLE : "Calcul Formel"

Géométrie analytique avec Maple : Etude de l'hyperbole équilatère

M.Gouy, G.Huvent, A.Ladureau

Notations utilisées

Un point est noté comme une liste à deux éléments, par exemple A:=[1,2]

```
> A:=[1,2]; B:=[5,7];
```

$$A := [1, 2]$$

$$B := [5, 7]$$

Cette instruction définie le point A de composantes [1,2] et B de composantes [5,7]..
On peut récupérer les coordonnées des points de la manière suivante :

```
> A[1];
```

$$1$$

Cette instruction donne la première composante de A.
On définit une droite par son équation, **attention, pour des raisons de simplicité, on ommet le "=0" dans l'équation des droites.**
Par exemple la droite d d'équation $4x - 3y = -2$ est définie par

```
> d:=4*x-3*y+2;
```

$$d := 4x - 3y + 2$$

De même, un cercle est défini par son équation normalisée

```
> cercle:=x^2+y^2-2*x-3;
```

$$cercle := x^2 + y^2 - 2x - 3$$

Procédures Maple

Pour faire l'étude proposée, on aura besoin de définir un certain nombre de **procédures.** Une procédure peut être vue comme un programme admettant des paramètres en entrée et donnant un résultat comme sortie.

Attention : Dans Maple, le résultat d'une procédure est le dernier calcul effectuée par cette dernière.
On rappelle que la syntaxe est la suivante (sur l'exemple de la procédure qui détermine le milieu de deux points)

```
> milieu:= proc(A,B);
  (A+B)/2;
  end;
```

$$milieu := \mathbf{proc}(A, B)\ 1\ /\ 2*A + 1\ /\ 2*B\ \mathbf{end\ proc}$$

```
> milieu(A,B);
```

$$\left[3, \frac{9}{2}\right]$$

Donnons un autre exemple, celui du calcul du produit scalaire des vecteurs OA et OB.

```
> scal:=proc(A,B);
  A[1]*B[1]+A[2]*B[2];
  end;
```

$$scal := \mathbf{proc}(A, B)\ A[1]*B[1] + A[2]*B[2]\ \mathbf{end\ proc}$$

```
> scal(A,B);
```

$$19$$

L'intérêt d'un logiciel de calcul formel réside dans la manipulation de données dans lesquelles figurent des paramètres.
Pour cela on re-définit A et B

```
> A:=[xa,ya];B:=[xb,yb];
```

$$A := [xa, ya]$$

$$B := [xb, yb]$$

```
> milieu(A,B);
```

$$\left[\frac{1}{2}xb + \frac{1}{2}xa, \frac{1}{2}yb + \frac{1}{2}ya\right]$$

```
> scal(A,B);
```

$$xa\,xb + ya\,yb$$

On peut aussi enchainer les procédures, par exemple le produit scalaire de OA et de OJ où J est le milieu de [A,B] sera calculé ainsi

```
> J:=(A+B)/2;
```

$$J := \left[\frac{1}{2}xb + \frac{1}{2}xa, \frac{1}{2}yb + \frac{1}{2}ya\right]$$

```
> scal(A,J);
```

$$xa\left(\frac{1}{2}xb + \frac{1}{2}xa\right) + ya\left(\frac{1}{2}yb + \frac{1}{2}ya\right)$$

Question : Si on définit `C:=[1,2]:` , comment définir le vecteur   AC

Quelques commandes utiles

La commande `normal:`

```
>
```

Elle permet de simplifier un résultat (elle met une expression sous forme normale, ne me demandez pas ce que Maple appelle forme normale, c'est assez compliqué et je n'ai pas de réponse exacte).
Par exemple.

```
> C:=[1,2];E:=[5,3];
```

$$C := [1, 2]$$

$$E := [5, 3]$$

```
> scal(C-A,E-B);
```

$$(-xa + 1)(-xb + 5) + (-ya + 2)(-yb + 3)$$

```
> normal(scal(C-A,E-B));
```

$$xa\,xb - 5\,xa - xb + 11 + ya\,yb - 3\,ya - 2\,yb$$

Cela incite à définir la procédure `scal:` ainsi

```
> scal:=proc(A,B);
  normal(A[1]*B[1]+A[2]*B[2]);
  end;
```

$$scal := \textbf{proc}(A, B)\ \mathrm{normal}(A[1]*B[1] + A[2]*B[2])\ \textbf{end proc}$$

La substitution : `subs:`

Comment vérifier que le point `C:` que l'on vient de définir appartient à la droite `d:` ?
Pour cela, on peut être tenté d'affecter à `x:` la valeur de `C[1]:`, et à `y:` celle de `C[2]:`. Il s'agit d'un mauvais choix. La bonne méthode consiste à effectuer une substitution (l'analogue du | x=c[1] des calculatrices TI)
La substitution est réalisée par la commande `subs:` Par exemple

```
> subs({x=C[1],y=C[2]},d);
```

$$0$$

effectue les susbtitutions `x=C[1]:` et `x=C[2]:` dans l'expression contenue dans `d:`.
(les {} permettent de définir un ensemble d'égalité)

```
>
```

Exercice : Le point `E:` est-il sur `d:` ?

La commande `solve:`

Elle permet la résolution d'un ensemble d'équations en fonction d'un ensemble d'inconnues.
Par exemple si l'on définit la droite

```
> d2:=x+a*y-2;
```

$$d2 := x + a\,y - 2$$

qui dépend du paramètre `a:`, le point d'intersection de `d:` et `d2:` sera déterminé ainsi

```
> solve({d,d2},{x,y});
```

$$\{x = -2\,\frac{-3 + a}{3 + 4\,a},\ y = 10\,\frac{1}{3 + 4\,a}\}$$

Si on désire définir le point `M:` comme intersection de `d:` et `d2:`, on utilisera

```
> M:=subs(solve({d,d2},{x,y}),[x,y]);
```

$$M := \left[-2\,\frac{-3 + a}{3 + 4\,a},\ 10\,\frac{1}{3 + 4\,a}\right]$$

Etude de l'hyperbole équilatère

Pour réaliser cette étude, on doit

- Définir quatre points A, B, C et M d'abscisses a, b, c, m sur l'hyperbole $y = \frac{1}{x}$
- Définir une procédure **droite:** dont les paramètres d'entrée sont : deux points A et B, et dont le paramètre de sortie est la droite (AB)
- Définir une procédure **intersection:** dont les paramètres d'entrée sont : deux droites d1 et d2, et dont le paramètre de sortie est le point d'intersection de d1 et d2.
- Définir une procédure **cerclecirc:** dont les paramètres d'entrée sont : trois points A,B,C et dont le paramètre de sortie est le cercle circonscrit.

On peut alors établir le théorème suivant :

> **Théorème :**
>
> Soit A,B,C et M quatre points de $y = \frac{1}{x}$. On note P l'intersection de AM et de BC, Q l'intersection de BM et de AC
> et R l'intersection de CM et de AB. Alors le cercle circonscrit au triangle cévien PQR passe par l'origine.

-Définir une procédure **hauteur:** dont les paramètres d'entrée sont : une droite d1 et un point A, et dont le paramètre de sortie est la droite perpendiculaire à d1 passant par A.

On peut alors établir le théorème suivant :

> **Théorème :**
>
> Soit A,B,C et M quatre points de $y = \frac{1}{x}$. On note S le projeté orthogonal de M sur AB, T celui sur BC et U celui sur AC. Alors Le cercle circonscrit au triangle pédal STU passe par l'origine.

On peut également considérer les cas de dégénérescence.

Si vous etes vraiment motivé, il vous reste à montrer le résultat suivant

> **Théorème :**
>
> Soit A,B,C et M quatre points de $y = \frac{1}{x}$. On note P l'intersection de AM et de BC, Q l'intersection de BM et de AC
> et R l'intersection de CM et de AB. Alors les centres du cercle inscrit et des cercles exinscrits au triangle cévien PQR sont aussi sur l'hyperbole.$y = \frac{1}{x}$

```
> A:=[a,1/a]; B:=[b,1/b]; C:=[c,1/c];
```

$$A := \left[a, \frac{1}{a}\right]$$

$$B := \left[b, \frac{1}{b}\right]$$

$$C := \left[c, \frac{1}{c}\right]$$

Procédure droite : paramètres d'entrée: deux points, paramètre de sortie : l'équation de la droite joigant ces deux points (équation du type ax+by+c)

```
>
```

```
> droite:=proc(A,B);
  normal((x-A[1])*(B[2]-A[2])-(y-A[2])*(B[1]-A[1]));
> end;
```

droite :=

proc(A, B) normal($(x - A[1])*(B[2] - A[2]) - (y - A[2])*(B[1] - A[1])$) **end proc**

La commande normal permet de simplifier le résultat

Procédure intersection : paramètres d'entrée: deux équation de droites , paramètre de sortie : le point d'intersection

```
> intersection:=proc(e1,e2) local s;
  s:=solve({e1,e2},{x,y});
  subs(s,[x,y])
  end;
```

intersection := **proc**($e1$, $e2$) **local** s; s := solve($\{e1, e2\}$, $\{x, y\}$); subs(s, $[x, y]$) **end proc**

Procédure hauteur : paramètres d'entrée: trois points A,B et C, paramètre de sortie : la hauteur issue de A

```
> hauteur:=proc(A,B,C);
  normal((x-A[1])*(C[1]-B[1])+(y-A[2])*(C[2]-B[2]));
  end;
```

hauteur :=

proc(A, B, C) normal($(x - A[1])*(C[1] - B[1]) + (y - A[2])*(C[2] - B[2])$) **end proc**

Procédure cercle_circons: paramètres d'entrée: trois points A,B et C, paramètre de sortie : l'équation du cercle circonscrit

Pour cela, on se souvient que l'équation d'un cercle est $x^2 + y^2 + a\,x + b\,y + c = 0$ Il suffit de trouver les coefficients a,b et c pour que le cercle passe par A,B et C

```
> cercle_circons:=proc(A,B,C)
  local a,b,c,eq,eq1,eq2,eq3;
  eq:=x^2+y^2+a*x+b*y+c;
  eq1:=subs({x=A[1],y=A[2]},eq);
  eq2:=subs({x=B[1],y=B[2]},eq);
  eq3:=subs({x=C[1],y=C[2]},eq);
```

```
subs(solve({eq1,eq2,eq3},{a,b,c}),eq);
end;
```

$cercle\_circons :=$ **proc**(A, B, C)
local $a, b, c, eq, eq1, eq2, eq3;$

$eq := x\verb|^|2 + y\verb|^|2 + a*x + b*y + c;$

$eq1 := \mathrm{subs}(\{x = A[1], y = A[2]\}, eq);$

$eq2 := \mathrm{subs}(\{x = B[1], y = B[2]\}, eq);$

$eq3 := \mathrm{subs}(\{x = C[1], y = C[2]\}, eq);$

$\mathrm{subs}(\mathrm{solve}(\{eq1, eq2, eq3\}, \{c, a, b\}), eq)$

end proc

Théorème : Soit M un point de y=1/x.
On note P l'intersection de AM et de BC
On note Q l'intersection de BM et de AC
On note R l'intersection de CM et de AB
Le cercle circonscrit à PQR passe par O

```
> M:=[m,1/m];
```

$$M := \left[m, \frac{1}{m}\right]$$

```
> P:=intersection(droite(A,M),droite(B,C));
```

$$P := \left[-\frac{a\,c\,b - m\,a\,c - m\,a\,b + m\,c\,b}{-c\,b + m\,a}, \frac{a - c + m - b}{-c\,b + m\,a}\right]$$

```
> Q:=intersection(droite(B,M),droite(A,C));
```

$$Q := \left[\frac{a\,c\,b - m\,c\,b - m\,a\,b + m\,a\,c}{c\,a - m\,b}, \frac{-b + c - m + a}{c\,a - m\,b}\right]$$

```
> R:=intersection(droite(C,M),droite(B,A));
```

$$R := \left[\frac{a\,c\,b + m\,a\,b - m\,c\,b - m\,a\,c}{-m\,c + a\,b}, \frac{-c - m + b + a}{-m\,c + a\,b}\right]$$

```
> subs({x=0,y=0},cercle_circons(P,Q,R));
```

$$0$$

Théorème : On note S le projeté orthogonal de M sur AB, T celui sur BC et U celui sur AC.
Le cercle circonscrit à STU passe par O

```
> S:=intersection(droite(A,B),hauteur(M,A,B));
```

$$S := \left[\frac{m\,a - a\,b + m^2\,a^2\,b^2 + m\,b}{m\,(1 + a^2\,b^2)}, \frac{m\,b\,a^2 + 1 + a\,m\,b^2 - a\,m^2\,b}{m\,(1 + a^2\,b^2)}\right]$$

```
> T:=intersection(droite(B,C),hauteur(M,B,C));
```

$$T := \left[\frac{m\,b + m\,c + m^2\,c^2\,b^2 - c\,b}{m\,(c^2\,b^2 + 1)}, \frac{m\,b^2\,c + m\,c^2\,b - m^2\,c\,b + 1}{m\,(c^2\,b^2 + 1)}\right]$$

```
> U:=intersection(droite(A,C),hauteur(M,A,C));
```

$$U := \left[\frac{m\,a - c\,a + m^2\,a^2\,c^2 + m\,c}{m\,(1 + c^2\,a^2)},\ \frac{a^2\,m\,c + 1 + a\,m\,c^2 - a\,m^2\,c}{m\,(1 + c^2\,a^2)}\right]$$

```
> subs({x=0,y=0},cercle_circons(S,T,U));
```

$$0$$

En toute rigueur, il faut examiner à part les cas où les cercles circonscrit n'existent pas.
Par exemple pour PQR

```
> cercle_circons(P,Q,R):normal(%):denom(%);
```

$$-2\,a^2\,b^2\,c^2 - 2\,m^2\,a^2\,b^2 - 2\,m^2\,a^2\,c^2 - 2\,m^2\,c^2\,b^2 + 2\,a\,m\,c\,b^3 + 2\,a\,b\,m\,c^3 + 2\,m\,c\,b\,a^3 + 2\,a\,c\,m^3\,b$$

```
> factor(%);
```

$$2\,(c\,a - m\,b)\,(-c\,b + m\,a)\,(-m\,c + a\,b)$$

```
> P;
```

$$\left[-\frac{a\,c\,b - m\,a\,c - m\,a\,b + m\,c\,b}{-c\,b + m\,a},\ \frac{a - c + m - b}{-c\,b + m\,a}\right]$$

```
> subs(m=c*b/a,droite(Q,R)):factor(%);
```

$$-\frac{(b - c)\,(-c\,b + a^2)\,(-y\,c\,b + x)}{(a + b)\,c\,(a + c)\,b}$$

Ce qui prouve que si P est à l'infini, la droite QR passe par l'origine.

```
>
> cercle_circons(S,T,U);
```

$$x^2 + y^2 - \frac{(a\,m^2\,c\,b - a - b - c)\,x}{m\,c\,b\,a - 1} - \frac{(a\,m^2\,b + a\,m^2\,c + m^2\,c\,b - 1)\,y}{(m\,c\,b\,a - 1)\,m}$$

Il y a un problème si $amcb = 1$

i.e. si M=F=symétrique de H par rapport à l'origine. Dans ce cas, les points S,T et U sont alignés sur une droite qui passe par O (le cercle circonscrit dégénère en une droite)

```
> cercl_degenSTU:=subs(m=1/a/b/c,droite(S,T));
```

$$cercl\_degenSTU := \left(-\frac{y}{a^2\,b^2\,c} + \frac{y}{a\,b^2\,c^2} - \frac{x\,b\,c}{a} + \frac{x\,a\,b}{c} + \frac{x}{b^2\,a^2\,c^3} - \frac{x}{b\,c^2} - y\,b^2\,c + \frac{x}{b\,a^2} - \frac{y\,b}{c^2} - \frac{y}{a^3\,b\,c} + \frac{y}{a\,b\,c^3} + \frac{y\,b}{a^2} - \frac{x}{b^2\,a^3\,c^2} + y\,a\,b^2 + \frac{2\,y}{a} + \frac{2\,x}{a^2\,c} + \frac{x\,b^2}{c} - \frac{2\,y}{c} - \frac{x\,b^2}{a} - \frac{2\,x}{a\,c^2}\right)a^2\,b^2\,c^2 \Big/ \left((c^2\,b^2 + 1)\,(1 + a^2\,b^2)\right)$$

```
> subs({x=0,y=0},cercl_degenSTU);
```

$$0$$

Ce qui prouve que O est sur la droite ST

```
> subs({x=U[1],y=U[2]},cercl_degen): subs(m=1/a/b/c,%):
  simplify(%);
```

$$0$$

Ce qui prouve que U est sur la droite ST

```
>
>
```